AD-781 759

XAREA--A FORTRAN SUBROUTINE TO FIND THE INTERSECTION AREA OF AN ANNULUS WITH A CIRCLE

William F. Peay, III.

Air Force Weapons Laboratory
Kirtland Air Force Base, New Mexico

June 1974

DISTRIBUTED BY:

NTIS

National Technical Information Service
U. S. DEPARTMENT OF COMMERCE
5285 Port Royal Road, Springfield Va. 22151

AFWL-TR-74-67

AIR FORCE WEAPONS LABORATORY
Air Force Systems Command
Kirtland Air Force Base
New Mexico 87117

ACCESSION for
NTIS White Section
Buff Section
BY
DISTRIBUTION/AVAILABILITY CODES
A

ic

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER AFWL-TR-74-67	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER AD-781 759
4. TITLE (and Subtitle) XAREA--A FORTRAN SUBROUTINE TO FIND THE INTERSECTION AREA OF AN ANNULUS WITH A CIRCLE		5. TYPE OF REPORT & PERIOD COVERED Final Report; 1 October 1972-1 February 1974
		6. PERFORMING ORG. REPORT NUMBER AFWL-TR-74-67
7. AUTHOR(s) William F. Peay III		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Air Force Weapons Laboratory (SAB) Kirtland Air Force Base, New Mexico 87117		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 62601F; 8809; 04
11. CONTROLLING OFFICE NAME AND ADDRESS		12. REPORT DATE June 1974
		13. NUMBER OF PAGES 36
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Annulus; Centroid; Circle area; Flush; Flush sequence; Function; Ground surface area; Intersection area; Probability of kill; Quanto; Segment area; Spherical destruction pattern; Spherical kill pattern; Subroutine; Weapon distribution; Weapon volume; XAREA

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

When aircraft are flushing from a base under attack, the portion of the force that will escape to carry out the mission must be known to a relatively high level of confidence. This involves the determination of the probability of kill associated with the base force. One major component of this term is the geometrical intersection of the weapon's effective kill volume with the region containing all aircraft at any instant in time. Subroutine XAREA calculates a numerical value for this intersection in terms of area according to the following assumptions: (1) all interaction takes place in a plane, parallel to the ground and

elevated to the average aircraft altitude; (2) the figure within which all aircraft are located is an annulus, bounded by the maximum effective ground distance reached by the first aircraft off the base and the minimum distance achieved by the last aircraft in the flush sequence; (3) the effective kill area of a weapon with an assumed spherical destruction pattern is the circle cut by the plane through the sphere. Although the intended use of this computer function is highly specialized, it is applicable under any conditions requiring the area of intersection of an annulus with a circle.

AFWL-TR-74-67

XAREA--A FORTRAN SUBROUTINE TO FIND THE INTERSECTION AREA OF AN ANNULUS WITH A CIRCLE

William F. Peay III

Final Report for Period 1 October 1972 through 1 February 1974

TECHNICAL REPORT NO. AFWL-TR-74-67

Approved for public release; distribution unlimited.

FOREWORD

The research was performed under Program Element 62601F, Project 8809, Task 04.

Inclusive dates of research were 1 October 1972 through 1 February 1974. The report was submitted 6 May 1974 by the Air Force Weapons Laboratory Project Officer, Major Arthur R. Geldbach (SAB).

The computer subroutine XAREA has been developed within the Air Force Weapons Laboratory to function as a part of the advanced computer model QUANTO--A CODE TO OPTIMIZE WEAPON ALLOCATIONS (AFWL-TR-73-242). The approach used in XAREA for providing the numerator portion of the P_k based on overlapping weapon-aircraft area has been reviewed by the Systems Analysis Section of the Battle Environments Branch, AFWL/SA, and is considered appropriate for use in activities relating to aircraft flush methods.

The basic model was developed upon an identification of need by Major Richard Conway. Assistance in the formulation of the unique branching mechanism was provided by Captain Karl T. Benson who, along with Major Arthur Geldbach, assured the completion of this report through their technical editing and experience.

This technical report has been reviewed and is approved.

Arthur R. Geldbach

ARTHUR R. GELDBACH
Major, USAF
Project Officer

George H. Dimon Jr.

GEORGE H. DIMON, JR.
Colonel, USAF
Chief, Battle Environments Branch

Charles C. Hyre Jr.

CHARLES C. HYRE, JR.
Colonel, USAF
Chief, Analysis Division

ABSTRACT

(Distribution Limitation Statement A)

When aircraft are flushing from a base under attack, the portion of the force that will escape to carry out the mission must be known to a relatively high level of confidence. This involves the determination of the probability of kill associated with the base force. One major component of this term is the geometrical intersection of the weapon's effective kill volume with the region containing all aircraft at any instant in time. Subroutine XAREA calculates a numerical value for this intersection in terms of area according to the following assumptions: (1) all interaction takes place in a plane, parallel to the ground and elevated to the average aircraft altitude; (2) the figure within which all aircraft are located is an annulus, bounded by the maximum effective ground distance reached by the first aircraft off the base and the minimum distance achieved by the last aircraft in the flush sequence; (3) the effective kill area of a weapon with an assumed spherical destruction pattern is the circle cut by the plane through the sphere. Although the intended use of this computer function is highly specialized, it is applicable under any conditions requiring the area of intersection of an annulus with a circle.

CONTENTS

ILLUSTRATIONS

ABBREVIATIONS AND SYMBOLS

P_k	The probability that an aircraft under attack will be destroyed or "killed."
XAREA	The intersection area of a specified annulus with a circle.
AH	The outer radius of an annulus.
AL	The inner radius of an annulus.
XL	The radius of a circle which intersects an annulus to produce XAREA.
XD	The displacement distance of the center of the annulus from the center of the intersecting circle.
XMIN	In a rectangular coordinate system oriented about the center of an annulus with its independent axis passing through the center of the intersecting circle, this is the horizontal (x) coordinate of the points generated from the intersection of the circle with the inner boundary of the annulus.
XMAX	In a rectangular coordinate system oriented about the center of an annulus with its independent axis passing through the center of the intersecting circle, this is the horizontal (x) coordinate of the points generated from the intersection of the circle with the outer boundary of the annulus.
ALANG	The angle formed between the line segments from (1) the center of the annulus to the center of the intersecting circle and (2) the center of the annulus to one of the inner annulus intersection points.
AHANG	The angle formed between the line segments from (1) the center of the annulus to the center of the intersecting circle and (2) the center of the annulus to one of the inner annulus intersection points.
XLANG	The angle formed between the line segments from (1) the center of the intersecting circle to the center of the annulus and (2) the center of the intersecting circle to the annulus intersection point (either inner or outer depending on where it is applied).
OPANG	The complementary angle opposite one of the above angles that is greater then 90 degrees when such a condition exists.

SECTION I

INTRODUCTION

The Air Force Weapons Laboratory (AFWL) has investigated weapon distributions and their relationship to kill probability, P_k, wherein attention was focused on the effectiveness of a weapon with a spherical destruction pattern detonated in a field of aircraft flying from an airbase under attack. These aircraft flushed single file and were commanded to fly at different headings by executing one single turn up to 180 degrees in any direction after achieving minimum required turn altitude and velocity. In this manner, they were constrained to a region bounded by the maximum distance reached by the first aircraft off the base and the minimum distance achieved by the last aircraft in the flush sequence. By examining all possible positions the first aircraft could reach in a given time interval for any turn angle, a two-dimensional figure can be generated representing the ground surface area over which all aircraft are located. As flight time progresses, this figure approximates a circle with the center, or centroid, located at some positive distance from the point of brake release in the direction of takeoff. Given enough time, the final aircraft in the flush sequence will escape from the base, leaving a void within which no aircraft are located. As flight time continues, this void region approximates a circle concentric with the maximum distance (i.e., first aircraft's) circle, but of smaller radius. Thus, after a period of elapsed time, the region containing all aircraft, referred to as the annulus, is the area inside a large circle and the area outside a smaller circle, both circles being centered about the same point. For simplicity, this annular region is assumed to be contained in a horizontal plane at some aircraft altitude. If a weapon with an effective kill volume is detonated at random in space, there will be no chance of destroying any of the aircraft unless the weapon volume intersects the region where the planes are located. Should this happen, then the kill probability can be computed based on the overlapping area. Assuming the weapon's lethal volume is spherical, then the lethal region in the horizontal plane containing the aircraft is a circle (figure 1). If the aircraft are uniformly distributed within the annular region, the probability of killing an aircraft is the quotient of (1) the area common to both the weapon's circular

lethal area and the aircrafts' annular region, divided by (2) the area of the annulus containing all aircraft. Subroutine XAREA has been created to compute the numerator of this quotient, for all combinations of values of the two radii of the annulus, the radius of the lethal circle of the weapon, and the offset distance of the weapon circle center form the center of the annulus.



Figure 1. Horizontal Annulus Plane Intersecting Lethal Weapon Volume

SECTION II

MATHEMATICAL DEVELOPMENT

Subroutine XAREA computes the area common to a circle and an annulus. The value returned to the main program is the area of intersection of the annular region of outer radius AH and inner radius AL, with a circular weapon area of lethal radius XL displaced a distance XD from the centroid. The centers of the circles defining the annulus are concentric with the centroid of the aircraft area. It should be noted that in all cases AH > AL and all measurements are taken to be nonnegative with respect to the centroid, in the same unit denominations.

The basic formulas used for the computation of the area of intersection involve repeated application of the rules governing the area of circles and segments of circles. The equations for these areas are

$$A\ (\text{circle}) = \pi R^2 \tag{1}$$

$$A\ (\text{segment}) = 1/2R^2(\theta - \sin\theta) \tag{2}$$

where R is the radius of the circle and θ is the central angle (in radians) of the segment. The circles are taken two at a time, and a triangle is constructed with vertices at the center of each circle and one of the two points where the circles intersect each other as shown in figure 2. In this example the three sides of the triangle are indicated as XD (distance between centers), XL (distance from weapon detonation point to circular intersection point), and AL (distance from centroid to circular intersection point). Any angle in a triangle may be found from the formula

$$\text{Angle} = \text{arc cos}\left[\frac{(\text{adjacent side})^2 + (\text{other adjacent side})^2 - (\text{opposite side})^2}{2 \cdot (\text{adjacent side}) \cdot (\text{other adjacent side})}\right] \tag{3}$$

By determining all the angles of the triangle AL-XL-XD and finding the area of all segments involved, a quick examination of the orientation of the intersection will reveal which segments should be added and which should be subtracted to obtain the correct solution.



Figure 2. Annulus-Circle Single Intersection

Assume that by using the above formulas the areas have been found for segments A and B in the orientation shown in figure 2. The area of intersection of the aircraft annulus with the weapon's lethal circle will clearly be the area of segment A minus the area of segment B. These areas arise from the application of formula (2), where XL represents R when calculating segment A, and AL represents R for the determination of segment B. Theta is angle β or α in figure 2, respectively, and is found from a direct application of formula (3) using the three triangle sides AL, XL, and XD. In the program itself these angles are named for quick identification. ALANG (AL angle) indicates the angle formed with AL and XD as adjacent sides (α in this case). XLANG (XL angle) is similar but has XL and XD as adjacent sides (β in figure 2). Both of these angles have XD as a common side. Note that the area of segment A does not follow directly from formula (2). The segment area found from using angle β and triangle side XL must be subtracted from the area of the lethal circle to obtain the actual area of segment A. This subtraction is not necessary for determining the area of segment B.

In this first example the angles of triangle AL-XL-XD are all acute angles. It is possible, however, for one of these angles to be larger than 90 degrees under certain geometrical conditions. The second example depicting this situation is identical to the first one with one alteration. The maximum distance the first aircraft off the base could achieve (AH) has been decreased so that the lethal circle entirely covers a portion of the annulus (figure 3). Although the computation of the area of segments A and B is identical to the previous example, two new segments, C and D, must be analyzed to determine the true intersectional area, for a portion of segment A furthest from the centroid extends beyond the intersection. Again, formula (2) is used to find these new segment areas. This time the radius R is represented by AH for segment C and XL for segment D. The angles μ and ψ represent θ in this equation, where ψ is derived by calculating the complimentary angle, ϕ, from formula (3) and subtracting the result from 180 degrees. In the program, angle μ is denoted as AHANG (AH angle), ψ is again XLANG, and ϕ is called OPANG (opposite angle) since it is complementary to angle ψ.



Figure 3. Annulus-Circle Double Intersection

SECTION III

THE PROGRAM

There are several conditions surrounding the intersection of a circle with an annulus. It is possible for the lethal circle to be smaller than the inner annulus circle, larger than the outer annulus circle, or between the two in relative size. Within this context, the entire lethal circle could be small enough to fit within the annulus, or large enough to overlap both the inner and outer dimensions of the annulus if properly oriented between the two. The same basic formulas are applicable in each case, but different approaches are required in summing the individual segment areas together properly to determine the actual area of intersection. It must be known, for example, if the lethal circle actually intersects the annulus. If it does, it should be determined if the intersection is with the inner annulus circle, the outer annulus circle, or both.

In all, there are seven different categories of tests required to isolate the geometrical orientation of a particular situation and define the proper mathematical scheme required for intersection area calculation. Within the program five of these categories have only one possible solution to a given series of questions presented as "IF" statements. A numerical value is assigned to each of the questions, ranging from one up to the total number of questions in the category. Only the numerical value for the correct solution to one question in each category is carried through the program, adjusted by a factor of 10 to distinguish the categories from each other. At the completion of the questioning period the five resultant numbers are summed up. Each digit of this final solution represents the correct answer from each category. This number is then compared against an array of possible number combinations until the mating value is located. At that point the control transfers to one of the solution blocks for actual intersection area computation through the use of a computed GO TO statement. If no matches have occurred, the subroutine will return an area of -1.0 as an error indicator for invalid input data since all proper solutions are nonnegative.

Appendix I has been provided to help trace through XAREA and find the section of code involved in calculating the intersection area for any specific example. By answering the questions in each test section using numerical values for the four input variables, the corresponding statement label of the start of the pertinent solution coding in XAREA can be identified. By following through this coding block in the program listing (appendix III), the area of intersection of the annulus with the lethal circle may be calculated for a specific situation. To implement subroutine XAREA on a computer system, a sample driver program has been provided in appendix II, with the expected output shown in appendix IV.

SECTION IV

SUMMARY

Although the formulas used for computing the area of circles and segments of circles are simple, there are several geometrical conditions surrounding the intersection of a circle with an annulus that require much more detailed analysis. In light of these complications, and with a strong desire to be able to calculate this area of intersection under any possible condition, a separate computer subroutine was developed as a subtask to a larger Air Force study. This function, called XAREA, has been written to satisfy a definite need recognized by the Air Froce Weapons Laboratory. It is, however, applicable to any situation requiring the area of intersection of an annulus with a circle.

APPENDIX I

CASE TEST AND EVALUATION

The following tests are designed to direct a user to the pertinent portion of the subroutine coding where the actual area calculation is performed for specific case situations. The input parameters required before beginning the tests are

- XL The lethal circle radius
- XD The offset distance of the lethal circle from the center of the annulus
- AL The radius of the inner annulus circle
- AH The radius of the outer annulus circle

The resultant coding location may be found in appendix III. The functional elements ANGLE, SEGMT1, and SEGMT2 are defined at the beginning of the code.

<u>TEST 1</u>

Size relationship of lethal circle to annulus. One of the following conditions will hold:

If $2XL \leq AH-AL$ and $XL \leq AL$, write down 10,000.

If $2XL \leq AH-AL$ and $AL < XL < AH$, write down 20,000.

If $2XL > AH-AL$ and $XL < AL$, write down 30,000.

If $2XL > AH-AL$ and $AL < XL \leq AH$, write down 40,000.

If $2XL > AH-AL$ and $XL > AH$, write down 50,000.

<u>TEST 2</u>

Relationship of left-most edge of the lethal circle to the annulus for positive values of XD measured to the right of the center of the annulus. One of the following conditions will hold:

If $XD-XL < -AH$, add 1000 to the number from test 1.

If $-AH \leq XD-XL \leq -AL$, add 2000 to the number from test 1.

If $-AL < XD-XL \leq AL$, add 3000 to the number from test 1.

If AL < XD-XL $\leq$ AH, add 4000 to the number from test 1.

If XD-XL > AH, add 5000 to the number from test 1.

TEST 3

Relationship of right-most edge of the lethal circle to the annulus for positive values of XD measured to the right of the center of the annulus. One of the following conditions will hold:

If XD+XL $\leq$ AL, add 100 to the number from test 2.

If AL < XD+XL $\leq$ AH, add 200 to the number from test 2.

If XD+XL > AH, add 300 to the number from test 2.

TEST 4

Determination of the intersection of the lethal circle with the outer annulus circle. An intersection exists if the number derived from tests 1, 2, and 3 matches any of the following:

14300	24300	33300	42300
		34300	43300
			44300

If the numbers match, then calculate XMAX, an indicator for obtuse angles in the triangle AH-XD-AL, from

$$XMAX = \frac{AH^2 + XD^2 - XL^2}{2 \cdot XD}$$

If the numbers do not match, set XMAX equal to a very large positive number approximating positive infinity. Due to computer limitation +1,000,000,000 was used. Note that the number combinations 52300, 53300, and 54300 also indicate an intersection of the lethal circle with the outer annulus circle; however, when XL > AH, this calculation of XMAX is not required.

TEST 5

Determination of the intersection of the lethal circle with the inner annulus circle. An intersection exists if the number derived from tests 1, 2, and 3 matches any of the following:

13200	23200	33200	43200	53300
		33300	43300	

If the numbers match, then calculate XMIN, an indicator for obtuse angles in the triangle AL-XD-XL, from

$$XMIN = \frac{AL^2 + XD^2 - XL^2}{2 \cdot XD}$$

If the numbers do not match, set XMIN equal to a very large negative number approximating negative infinity. Due to computer limitation -1,000,000,000 was used.

TEST 6

Branching for special obtuse angle considerations through the relationship of the center of the lethal circle to the intersection points found in tests 4 and 5. One of the following conditions will hold:

If XD $\leq$ XMIN, add 10 to the number from test 3.

If XMIN < XD $\leq$ XMAX, add 20 to the number from test 3.

If XD > XMAX, add 30 to the number from test 3.

TEST 7

Orientation of the intersection of the lethal circle with the inner annulus circle in relation to the center of the annulus. One of the following conditions will hold:

If XMIN $\leq$ 0, add 1 to the previous number.

If XMIN > 0, add 2 to the previous number.

TEST EVALUATION

The five digit number calculated from the test section indicates which set of mathematical functions coded in XAREA are used to calculate the actual intersection area for the geometrical orientation under evaluation. Match this number with the exact same value listed below to determine the label location used in the code for area calculation. If none of the numbers match, an error has been made in either the input data or the test section.

Number	Label	Number	Label	Number	Label
13121	3	33121	3	43321	16
13212	4	33212	4	43322	12
13222	5	33222	5	43331	17
14221	6	33312	11	43332	13
14321	7	33322	12	44321	7
14331	8	33332	13	44331	3
15321	3	34321	7	45321	3
22221	9	34331	8	51321	20
23221	10	35321	3	52321	15
23222	5	42221	9	53312	22
24221	6	42321	14	53321	21
24321	7	42331	15	53322	22
24331	8	43221	10	54321	8
25321	3	43222	5	55321	3

APPENDIX II

DRIVER PROGRAM LISTING

```
      PROGRAM DRIVER(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      M=5
      N=6
      XLINES=50.
    1 READ(M,2) XL,AL,AH,XD
    2 FORMAT(4F10.5)
      IF(XL.LT.0.0)STOP
      WRITE(N,3) XL,AL,AH
    3 FORMAT(1H1,10X,3HXL=,F10.5,5X,3HAL=,F10.5,5X,3HAH=,F10.5//
     116X,2HXD,10X,17HINTERSECTING AREA,12X,2HPK/)
    4 AREA=XAREA(XL,XD,AL,AH)
      TAREA=3.141592654*(AH**2-AL**2)
      PK=AREA/TAREA
      WRITE(N,5) XD,AREA,PK
    5 FORMAT(10X,F10.5,10X,F12.5,10X,F10.7)
      XD=XD+(AH+XL)/XLINES
      IF(XD.LT.AH+XL) GO TO 4
      GO TO 1
      END
```

APPENDIX III

SUBFUNCTION LISTING

```
      FUNCTION XAREA (XL,XD,AL,AH)                                      WFP00100
      DIMENSION NUMBER(17), MUMBER(6)                                   WFP00110
      DATA PI/3.1415926536/                                             WFP00120
      DATA (NUMBER(J),J=1,17)/2121,2221,2321,2331,3121,3212,3221,3222,33WFP00130
     112,3321,3322,3331,3332,4221,4321,4331,5321/                       WFP00140
      DATA (MUMBER(J),J=1,6)/51,61,71,72,81,91/                         WFP00150
C                                                                       WFP00160
C                                                                       WFP00170
C     SUBROUTINE XAREA RETURNS THE INTERSECTION                         WFP00180
C     AREA OF AN ANNULUS WITH A CIRCLE.                                 WFP00190
C                                                                       WFP00200
C                                                                       WFP00210
C     INPUT VARIABLE DESCRIPTION-                                       WFP00220
C                                                                       WFP00230
C       XL -- THE RADIUS OF THE INTERSECTING CIRCLE                     WFP00240
C       XD -- THE POSITIVE DISTANCE BETWEEN THE                         WFP00250
C             CENTER OF THE INTERSECTING CIRCLE                         WFP00260
C             AND THE CENTER OF THE ANNULUS                             WFP00270
C       AL -- THE OUTER RADIUS OF THE ANNULUS                           WFP00280
C       AH -- THE INNER RADIUS OF THE ANNULUS                           WFP00290
C                                                                       WFP00300
C                                                                       WFP00310
C     RESTRICTIONS-                                                     WFP00320
C                                                                       WFP00330
C       1.  ALL NUMBERS MUST BE REAL AND NON-NEGATIVE.                  WFP00340
C       2.  AH MUST BE GREATER THEN AL.                                 WFP00350
C       3.  ALL NUMBERS MUST BE IN THE SAME UNITS.                      WFP00360
C                                                                       WFP00370
C                                                                       WFP00380
C     **********************************************************        WFP00390
C                                                                       WFP00400
C     ROUTINE DEVELOPED BY BILL PEAY, AFWL/SAB, KAFB, NEW MEXICO        WFP00410
C                         12 OCTOBER 1972                               WFP00420
C                  REVISED 24 JANUARY 1974 (WFP)                        WFP00430
C                                                                       WFP00440
C     **********************************************************        WFP00450
C                                                                       WFP00460
C                                   DEFINED FUNCTIONS--                 WFP00470
C                                                                       WFP00480
      ANGLE(AA,BB,CC)=ACOS((AA**2+BB**2-CC**2)/(2.0*AA*BB))             WFP00490
      SEGMT1(R,THETA)=R**2*(PI-(THETA-SIN(THETA))/2.0)                  WFP00500
      SEGMT2(R,THETA)=(R**2*(THETA-SIN(THETA)))/2.0                     WFP00510
C                                                                       WFP00520
C                                   SET XMAX  MIN REFERENCE LEVELS      WFP00530
      XMAX=+1.E+9                                                       WFP00540
      XMIN=-1.E+9                                                       WFP00550
```

```
C                                   PRELIMINARY BRANCHING
      IF (AL.GE.AH) GO TO 2
      IF (XL.GT.AH) GO TO 18
C                                   MAIN TEST BLOCK
C                                   TEST FOR LEFT INTERSECTION
      A=XD-XL
      IF (A.GE.(-AH).AND.A.LE.(-AL)) I4=2000
      IF (A.GT.(-AL).AND.A.LE.(AL)) I4=3000
      IF (A.GT.AL.AND.A.LE.AH) I4=4000
      IF (A.GT.AH) I4=5000
C                                   TEST FOR RIGHT INTERSECTION
      B=XD+XL
      IF (B.LE.AL) I3=100
      IF (B.GT.AL.AND.B.LE.AH) I3=200
      IF (B.GT.AH) I3=300
C                                   TESTS INVOLVING XMAX AND XMIN
C
      IF (I4.EQ.3000.AND.I3.GE.200) XMIN=(XD**2+AL**2-XL**2)/(2.0*XD)
      IF (I3.EQ.300.AND.I4.LE.4000) XMAX=(XD**2+AH**2-XL**2)/(2.0*XD)
      IF (XD.LE.XMIN) I2=10
      IF (XD.GT.XMIN.AND.XD.LE.XMAX) I2=20
      IF (XD.GT.XMAX) I2=30
      IF (XMIN.LE.0.0) I1=1
      IF (XMIN.GT.0.0) I1=2
C                                   DETERMINE COMPONENT SUM
      NUM=I4+I3+I2+I1
C                                   CALCULATE ROUTING ARGUMENT
      ITST=1
      DO 1 J=1,17
      IF (NUM.GE.NUMBER(J)) ITST=ITST+1
1     CONTINUE
C                                   BRANCH TO APPROPRIATE CODE BLOCK
C
      GO TO (2,3,9,14,15,3,4,10,5,11,16,12,17,13,6,7,8,3), ITST
C
C                                   CODE BLOCK SECTION--
2     XAREA=-1.0
      RETURN
3     XAREA=0.0
      RETURN
4     ALANG=ANGLE(AL,XD,XL)*2.0
      OPANG=ANGLE(AL,XL,XD)
      XLANG=(ALANG/2.0+OPANG)*2.0
      ASEGMT=SEGMT2(XL,XLANG)
      BSEGMT=SEGMT2(AL,ALANG)
      XAREA=ASEGMT-BSEGMT
      RETURN
5     ALANG=ANGLE(AL,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AL)*2.0
      ASEGMT=SEGMT1(XL,XLANG)
      BSEGMT=SEGMT2(AL,ALANG)
      XAREA=ASEGMT-BSEGMT
      RETURN
```

```
6       XAREA=PI*XL**2
        RETURN
7       AHANG=ANGLE(AH,XD,XL)*2.0
        OPANG=ANGLE(AH,XL,XD)
        XLANG=(AHANG/2.0+OPANG)*2.0
        ASEGMT=SEGMT1(XL,XLANG)
        BSEGMT=SEGMT2(AH,AHANG)
        XAREA=ASEGMT+BSEGMT
        RETURN
8       AHANG=ANGLE(AH,XD,XL)*2.0
        XLANG=ANGLE(XL,XD,AH)*2.0
        ASEGMT=SEGMT2(XL,XLANG)
        BSEGMT=SEGMT2(AH,AHANG)
        XAREA=ASEGMT+BSEGMT
        RETURN
9       XAREA=PI*(XL**2-AL**2)
        RETURN
10      XLANG=ANGLE(XL,XD,AL)*2.0
        OPANG=ANGLE(XL,AL,XD)
        ALANG=(XLANG/2.0+OPANG)*2.0
        ASEGMT=SEGMT2(AL,ALANG)
        BSEGMT=SEGMT2(XL,XLANG)
        XAREA=PI*(XL**2-AL**2)+ASEGMT-BSEGMT
        RETURN
11      ALANG=ANGLE(AL,XD,XL)*2.0
        OPANG=ANGLE(AL,XL,XD)
        XLANG=(ALANG/2.0+OPANG)*2.0
        ASEGMT=SEGMT2(XL,XLANG)
        BSEGMT=SEGMT2(AL,ALANG)
        AHANG=ANGLE(AH,XD,XL)*2.0
        OPANG=ANGLE(AH,XL,XD)
        XLANG=(AHANG/2.0+OPANG)*2.0
        CSEGMT=SEGMT2(XL,XLANG)
        DSEGMT=SEGMT2(AH,AHANG)
        XAREA=ASEGMT+DSEGMT-BSEGMT-CSEGMT
        RETURN
12      ALANG=ANGLE(AL,XD,XL)*2.0
        XLANG=ANGLE(XL,XD,AL)*2.0
        ASEGMT=SEGMT1(XL,XLANG)
        BSEGMT=SEGMT2(AL,ALANG)
        AHANG=ANGLE(AH,XD,XL)*2.0
        OPANG=ANGLE(AH,XL,XD)
        XLANG=(AHANG/2.0+OPANG)*2.0
        CSEGMT=SEGMT2(XL,XLANG)
        DSEGMT=SEGMT2(AH,AHANG)
        XAREA=ASEGMT+DSEGMT-BSEGMT-CSEGMT
        RETURN
```

```
13    ALANG=ANGLE(AL,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AL)*2.0
      ASEGMT=SEGMT2(XL,XLANG)
      BSEGMT=SEGMT2(AL,ALANG)
      AHANG=ANGLE(AH,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AH)*2.0
      CSEGMT=SEGMT2(XL,XLANG)
      DSEGMT=SEGMT2(AH,AHANG)
      XAREA=CSEGMT+DSEGMT-ASEGMT-BSEGMT
      RETURN
14    AHANG=ANGLE(AH,XD,XL)*2.0
      OPANG=ANGLE(AH,XL,XD)
      XLANG=(AHANG/2.0+OPANG)*2.0
      ASEGMT=SEGMT1(XL,XLANG)
      BSEGMT=SEGMT2(AH,AHANG)
      XAREA=ASEGMT+BSEGMT-PI*AL**2
      RETURN
15    AHANG=ANGLE(AH,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AH)*2.0
      ASEGMT=SEGMT2(XL,XLANG)
      BSEGMT=SEGMT2(AH,AHANG)
      XAREA=ASEGMT+BSEGMT-PI*AL**2
      RETURN
16    XLANG=ANGLE(XL,XD,AL)*2.0
      OPANG=ANGLE(XL,AL,XD)
      ALANG=(XLANG/2.0+OPANG)*2.0
      ASEGMT=SEGMT2(AL,ALANG)
      BSEGMT=SEGMT2(XL,XLANG)
      AHANG=ANGLE(AH,XD,XL)*2.0
      OPANG=ANGLE(AH,XL,XD)
      XLANG=(AHANG/2.0+OPANG)*2.0
      CSEGMT=SEGMT2(XL,XLANG)
      DSEGMT=SEGMT2(AH,AHANG)
      XAREA=PI*(XL**2-AL**2)+ASEGMT+DSEGMT-BSEGMT-CSEGMT
      RETURN
17    XLANG=ANGLE(XL,XD,AL)*2.0
      OPANG=ANGLE(XL,AL,XD)
      ALANG=(XLANG/2.0+OPANG)*2.0
      ASEGMT=SEGMT2(AL,ALANG)
      BSEGMT=SEGMT2(XL,XLANG)
      AHANG=ANGLE(AH,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AH)*2.0
      CSEGMT=SEGMT2(XL,XLANG)
      DSEGMT=SEGMT2(AH,AHANG)
      XAREA=ASEGMT+CSEGMT+DSEGMT-BSEGMT-PI*AL**2
      RETURN
```

```
C
C                                    SPECIAL TEST BLOCK FOR XL.GT.AH
C                                    TEST FOR LEFT INTERSECTION
18    A=XD-XL
      IF (A.LE.(-AH)) I4=50
      IF (A.GT.(-AH).AND.A.LE.(-AL)) I4=60
      IF (A.GT.(-AL).AND.A.LE.(AL)) I4=70
      IF (A.GT.AL.AND.A.LE.AH) I4=80
      IF (A.GT.AH) I4=90
C                                    TESTS INVOLVING XMIN
C
      IF (I4.EQ.700) XMIN=(XD**2+XL**2-AL**2)/(2.0*XD)
      IF (XMIN.LE.0.0) I3=1
      IF (XMIN.GT.0.0) I3=2
C                                    DETERMINE COMPONENT SUM
      NUM=I4+I3
C                                    CALCULATE ROUTING ARGUMENT
      ITST=1
      DO 19 J=1,6
      IF (NUM.GE.NUMBER(J)) ITST=ITST+1
19    CONTINUE
C                                    BRANCH TO APPROPRIATE CODE BLOCK
C
      GO TO (2,20,15,21,22,8,3), ITST
C
C                                    CODE BLOCK SECTION--
20    XAREA=PI*(AH**2-AL**2)
      RETURN
21    AHANG=ANGLE(AH,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AH)*2.0
      ASEGMT=SEGMT2(XL,XLANG)
      BSEGMT=SEGMT2(AH,AHANG)
      XLANG=ANGLE(XL,XD,AL)*2.0
      OPANG=ANGLE(XL,AL,XD)
      ALANG=(XLANG/2.0+OPANG)*2.0
      CSEGMT=SEGMT2(XL,XLANG)
      DSEGMT=SEGMT2(AL,ALANG)
      XAREA=ASEGMT+BSEGMT+DSEGMT-CSEGMT-PI*AL**2
      RETURN
22    AHANG=ANGLE(AH,XD,XL)*2.0
      XLANG=ANGLE(XL,XD,AH)*2.0
      ASEGMT=SEGMT2(XL,XLANG)
      BSEGMT=SEGMT2(AH,AHANG)
      XLANG=ANGLE(XL,XD,AL)*2.0
      ALANG=ANGLE(AL,XD,XL)*2.0
      CSEGMT=SEGMT2(AL,ALANG)
      DSEGMT=SEGMT2(XL,XLANG)
      XAREA=ASEGMT+BSEGMT-CSEGMT-DSEGMT
      RETURN
      END
```

APPENDIX IV

SAMPLE OUTPUT

XL= 2.00000 AL= 3.00000 AH= 10.00000

XD	INTERSECTING AREA	PK
0.00000	0.00000	0.0000000
.24000	0.00000	0.0000000
.48000	0.00000	0.0000000
.72000	0.00000	0.0000000
.96000	0.00000	0.0000000
1.20000	.37834	.0013234
1.44000	1.12728	.0039431
1.68000	1.99984	.0069953
1.92000	2.92944	.0102469
2.16000	3.88358	.0135844
2.40000	4.84280	.0169397
2.64000	5.79372	.0202659
2.88000	6.72599	.0235269
3.12000	7.63074	.0266917
3.36000	8.49974	.0297313
3.60000	9.32468	.0326169
3.84000	10.09655	.0353168
4.08000	10.80490	.0377946
4.32000	11.43653	.0400040
4.56000	11.97297	.0418804
4.80000	12.38292	.0433144
5.04000	12.56637	.0439560
5.28000	12.56637	.0439560
5.52000	12.56637	.0439560
5.76000	12.56637	.0439560
6.00000	12.56637	.0439560
6.24000	12.56637	.0439560
6.48000	12.56637	.0439560
6.72000	12.56637	.0439560
6.96000	12.56637	.0439560
7.20000	12.56637	.0439560
7.44000	12.56637	.0439560
7.68000	12.56637	.0439560
7.92000	12.56637	.0439560
8.16000	12.37909	.0433010
8.40000	11.84675	.0414389
8.64000	11.15128	.0390062
8.88000	10.35072	.0362059
9.12000	9.47901	.0331567
9.36000	8.56029	.0299431
9.60000	7.61358	.0266316
9.84000	6.65491	.0232783
10.08000	5.69857	.0199331
10.32000	4.75794	.0166428
10.56000	3.84615	.0134535
10.80000	2.97680	.0104126
11.04000	2.16476	.0075721
11.28000	1.42762	.0049937
11.52000	.78835	.0027576
11.76000	.28254	.0009883
12.00000	.00000	.0000000

XL= 5.00000 AL= 9.00000 AH= 10.00000

XD	INTERSECTING AREA	PK
0.00000	0.00000	0.0000000
.30000	0.00000	0.0000000
.60000	0.00000	0.0000000
.90000	0.00000	0.0000000
1.20000	0.00000	0.0000000
1.50000	0.00000	0.0000000
1.80000	0.00000	0.0000000
2.10000	0.00000	0.0000000
2.40000	0.00000	0.0000000
2.70000	0.00000	0.0000000
3.00000	0.00000	0.0000000
3.30000	0.00000	0.0000000
3.60000	0.00000	0.0000000
3.90000	0.00000	0.0000000
4.20000	.55188	.0092457
4.50000	2.10544	.0352727
4.80000	4.11938	.0690126
5.10000	6.24355	.1045991
5.40000	7.50160	.1256755
5.70000	8.36584	.1401542
6.00000	9.00187	.1508097
6.30000	9.47963	.1588136
6.60000	9.83843	.1648247
6.90000	10.10337	.1692633
7.20000	10.29168	.1724181
7.50000	10.41576	.1744967
7.80000	10.48483	.1756540
8.10000	10.50593	.1760075
8.40000	10.48450	.1756483
8.70000	10.42474	.1746473
9.00000	10.32996	.1730594
9.30000	10.20269	.1709273
9.60000	10.04485	.1682829
9.90000	9.85779	.1651491
10.20000	9.64238	.1615403
10.50000	9.39901	.1574631
10.80000	9.12761	.1529162
11.10000	8.82755	.1478893
11.40000	8.49765	.1423624
11.70000	8.13599	.1363035
12.00000	7.73970	.1296644
12.30000	7.30463	.1223755
12.60000	6.82466	.1143346
12.90000	6.29061	.1053875
13.20000	5.68775	.0952877
13.50000	4.98997	.0835977
13.80000	4.13992	.0693567
14.10000	2.89877	.0485635
14.40000	1.58538	.0265601
14.70000	.56312	.0094340
15.00000	.00000	.0000000

XL= 5.00000 AL= 1.00000 AH= 10.00000

XD	INTERSECTING AREA	PK
0.00000	75.39822	.2424242
.30000	75.39822	.2424242
.60000	75.39822	.2424242
.90000	75.39822	.2424242
1.20000	75.39822	.2424242
1.50000	75.39822	.2424242
1.80000	75.39822	.2424242
2.10000	75.39822	.2424242
2.40000	75.39822	.2424242
2.70000	75.39822	.2424242
3.00000	75.39822	.2424242
3.30000	75.39822	.2424242
3.60000	75.39822	.2424242
3.90000	75.39822	.2424242
4.20000	75.57813	.2430027
4.50000	76.05929	.2445497
4.80000	76.63642	.2464054
5.10000	77.04684	.2477250
5.40000	76.34759	.2454767
5.70000	75.01119	.2411798
6.00000	73.13757	.2351557
6.30000	70.75032	.2274801
6.60000	68.18559	.2192338
6.90000	65.48726	.2105580
7.20000	62.68811	.2015580
7.50000	59.81375	.1923162
7.80000	56.88488	.1828992
8.10000	53.91870	.1733622
8.40000	50.92985	.1637523
8.70000	47.93106	.1541104
9.00000	44.93360	.1444728
9.30000	41.94765	.1348722
9.60000	38.98257	.1253388
9.90000	36.04712	.1159005
10.20000	33.14960	.1065843
10.50000	30.29808	.0974159
10.80000	27.50045	.0884208
11.10000	24.76462	.0796245
11.40000	22.09862	.0710526
11.70000	19.51077	.0627320
12.00000	17.00980	.0546908
12.30000	14.60510	.0469591
12.60000	12.30695	.0395699
12.90000	10.12686	.0325604
13.20000	8.07809	.0259731
13.50000	6.17642	.0198587
13.80000	4.44141	.0142802
14.10000	2.89877	.0093203
14.40000	1.58538	.0050974
14.70000	.56312	.0018106
15.00000	.00000	.0000000

XL= 8.00000 AL= 7.00000 AH= 10.00000

XD	INTERSECTING AREA	PK
0.00000	47.12389	.2941176
.36000	47.12389	.2941176
.72000	47.12389	.2941176
1.08000	47.43221	.2960420
1.44000	50.60648	.3158538
1.80000	54.81001	.3420895
2.16000	58.68561	.3662787
2.52000	60.22305	.3758743
2.88000	60.86795	.3798994
3.24000	61.03798	.3809607
3.60000	60.90900	.3801557
3.96000	60.57476	.3780695
4.32000	60.09119	.3750514
4.68000	59.49415	.3713250
5.04000	58.80770	.3670406
5.40000	58.04861	.3623029
5.76000	57.22883	.3571863
6.12000	56.35703	.3517451
6.48000	55.43959	.3460190
6.84000	54.48115	.3400371
7.20000	53.48512	.3338205
7.56000	52.45390	.3273842
7.92000	51.38911	.3207385
8.28000	50.29170	.3138891
8.64000	49.16206	.3068386
9.00000	48.00011	.2995864
9.36000	46.80524	.2921289
9.72000	45.57643	.2844594
10.08000	44.31215	.2765685
10.44000	43.01037	.2684436
10.80000	41.66848	.2600684
11.16000	40.28320	.2514224
11.52000	38.85046	.2424801
11.88000	37.36514	.2332097
12.24000	35.82086	.2235712
12.60000	34.20945	.2135138
12.96000	32.52032	.2029714
13.32000	30.73930	.1918554
13.68000	28.84664	.1800425
14.04000	26.81302	.1673500
14.40000	24.59041	.1534779
14.76000	22.08238	.1378243
15.12000	18.92856	.1181402
15.48000	15.54504	.0970224
15.84000	12.37714	.0772503
16.20000	9.44677	.0589608
16.56000	6.78173	.0423273
16.92000	4.41918	.0275817
17.28000	2.41325	.0150620
17.64000	.85594	.0053422
18.00000	.00000	.0000000

XL= 10.00000 AL= 1.00000 AH= 9.00000

XD	INTERSECTING AREA	PK
0.00000	251.32741	1.0000000
.38000	251.32741	1.0000000
.76000	251.32741	1.0000000
1.14000	250.44449	.9964870
1.52000	245.79353	.9779814
1.90000	239.98318	.9548627
2.28000	233.69050	.9298250
2.66000	227.14894	.9037969
3.04000	220.46681	.8772095
3.42000	213.70291	.8502969
3.80000	206.89277	.8232002
4.18000	200.05973	.7960124
4.56000	193.22024	.7687989
4.94000	186.38658	.7416086
5.32000	179.56842	.7144800
5.70000	172.77372	.6874448
6.08000	166.00931	.6605300
6.46000	159.28123	.6337599
6.84000	152.59502	.6071563
7.22000	145.95583	.5807398
7.60000	139.36856	.5545299
7.98000	132.83798	.5285455
8.36000	126.36876	.5028053
8.74000	119.96554	.4773277
9.12000	113.71375	.4524526
9.50000	108.01243	.4297678
9.88000	102.56294	.4080850
10.26000	97.22090	.3868297
10.64000	91.87607	.3655633
11.02000	86.33947	.3435338
11.40000	80.53349	.3204326
11.78000	74.83394	.2977548
12.16000	69.24679	.2755242
12.54000	63.77836	.2537660
12.92000	58.43537	.2325070
13.30000	53.22502	.2117756
13.68000	48.15502	.1916027
14.06000	43.23375	.1720216
14.44000	38.47036	.1530687
14.82000	33.87492	.1347840
15.20000	29.45865	.1172123
15.58000	25.23417	.1004036
15.96000	21.21592	.0844155
16.34000	17.42067	.0693147
16.72000	13.86837	.0551805
17.10000	10.58337	.0421099
17.48000	7.59660	.0302259
17.86000	4.94950	.0196934
18.24000	2.70250	.0107529
18.62000	.95841	.0038134
19.00000	.00000	.0000000

```
 02/25/74  SCOPE 3.2.0 - SCM  VER 130 43   08FEB74ECPA= 1730
08.16.22.WFPNE73 AD
08.16.22.WFPNEW,P6,T40,CM60000.
08.16.22.TASK(LORING,88090904-9EM,SAB,2295)
08.16.22.FTN(B=FTNXAR,D)
08.16.53.     3.485 CP SECONDS COMPILATION TIME
08.16.54.MAP(OFF)
08.16.54.REWIND(OUTPUT)
08.16.54.PRESET.
08.16.55.FTNXAR(LC=377777)
08.17.00. 15200 CM
08.17.02.STOP
08.17.02.CP  004.255 SEC.
08.17.02.PP  034.971 SEC.
```